
ABSTRACT  

In today’s workplace, project teams are often virtual and spread across the country (or even the world). It 
is never a pleasant surprise to receive a meeting invitation two hours outside one’s working hours. When 
planning for meetings and events, it is important for the organizers to know where and in what time zone 
the attendees are located. Along with many other mapping tools, SAS® provides time zone information by 
zip code that can be used to map time zones and identify time for employee locations. This presentation 
will provide examples of mapping time zones in SAS. 

Examples in this presentation are created with SAS 9.4 and SAS Enterprise Guide 7.1 running in a grid 
environment and can be adapted by users of all levels. 

INTRODUCTION  

Before the advent of clocks, time of day was dependent upon the sun and time devices such as sun-dials. 
Mechanical clocks began to replace sun dials early in the 19th century but were based on local solar time, 
so time would be different in each locality. Greenwich Mean Time (GMT) was established in 1675 when 
the Royal Observatory was built to determine longitude at sea. This became one of the first standard 
reference times while each city in England kept different local times. 

The first person known to propose a worldwide system of time zones was Italian mathematician Quirico 
Filopanti in his 1858 book Miranda!, but there is no evidence that his idea influenced the actual adoption 
of time zones. He did propose 24 hourly time zones called “longitudinal days”, centered on the meridian 
of Rome. Scottish-born Canadian Sir Sandford Fleming is widely credited with the invention of time 
zones. He proposed a worldwide system of time zones in 1879 and advocated his system at several 
international conferences. He suggested that a universal day would begin at the anti-meridian of 
Greenwich (180th meridian), while conceding that hourly time zones might have some limited local use. 
He again pushed his system at the International Meridian Conference in 1884, but it did not adopt his time 
zones. That conference did adopt a universal day of 24 hours beginning at Greenwich midnight, but didn't 
want to interfere with local times. 

Today the earth is made up of 24 main time zones on Earth that compute their local time as an offset 
from GMT (also known as UTC or Universal Time Coordinated), each time zone boundary being 15 
degrees east or west of the preceding one. The reference point for GMT or UTC is the Greenwich 
Meridian (the Prime Meridian), which has a longitude of 0°. Each area that has a uniform, legally 
mandated standard time, referred to as the local time. Generally, local time is UTC plus the current time 
zone offset for the location. Time increases eastward from the UTC time zone by one hour for each 15°, 
up to the International Date Line (longitude 180°). A corresponding one hour decrease up to the 
International Date Line occurs for the westward direction.  

By about 1900, most of the Earth used standard time zones to denote time, but the hourly offsets from 
GMT was not yet in widespread use. It took until about 1930 before most major countries had adopted a 
“standard offset” from GMT. Nepal was the last country to adopt a standard offset from GMT, not doing so 
until 1986. While all nations today used standard time zones, there is much variation in how the concept 
is applied with half and quarter hour variations common. 

PROBLEM DESCRIPTION  

The number of off-premises workers in American businesses has been steadily increasing. There are a 
number of reasons why this is occurring. Employees are available over a greater number of hours, the 
company requires less infrastructure so saves costs, employees are happier, the talent pool is not 

restricted by geographic boundaries, and, in this era of thinking 'green', companies can have a lighter 
carbon footprint. 

Real-time communication is needed for meetings and work-groups. These meetings, while often virtual, 
make employees much happier if they occur during all participants' regular working hours. When 
scheduling a meeting, it is important to know where the participants are located and what time it is where 
they are located.  

The map below shows the location of the members of my team at the time of writing this paper. This map 
provides the first element - location, but not the second - time zone information. 

 

Figure 1. Map of Team Location.  

Here is the code to produce the above map where team_loc is a data set with counts of team members 
by state location. The data set also includes city information for each member. This information will be 
used in the time zone maps that are created in this paper. 

/*Set mapping colors*/ 

GOPTIONS colors= (cxBDD7E7, cx6BAED6, cx2171B5);  

 

/*Create the map with proc gmap*/ 

Proc GMAP map=mapsgfk.us data=team_loc; 

choro Members; 

id statecode; 

run; quit; 

NOTE: The map above points to the SAS MAPSGFK map library. SAS provides two types of SAS 
datasets for mapping. Traditional map datasets have been available with SAS/GRAPH since the product 
was released. Since SAS 9.3, digital, vector-based base maps supplied by GfK Geomarketing are also 

available for use. Gfk means ‘Growth from Knowledge’ and GFK is currently the fourth largest market 
research organization in the world. While both types of maps continue to be updated, the GfK maps are 

intended to eventually replace the traditional maps and at some point in the future the traditional maps will 
no longer be updated. Advantages include a single source for map data throughout the world and 
increased resolution for many boundaries. It is important to use the correct SAS map library to access 
these map datasets.  

 MAPSGFK points to the GfK Geomarketing map datasets. 

 MAPSSAS points to the traditional SAS boundary file. 

 Unless updated to point to the GfK maps, MAPS refers to the MAPSSAS datasets. 

SOLUTION ONE: COLOR-CODED TIME ZONES  

Because I had not created a time zone map and expected that someone had, I first reached out to my 
colleagues at https://communities.sas.com. I was quickly pointed to Bob Allison's wonderful page of SAS 
graphics examples, http://robslink.com/SAS/Home.htm, where you can find all sorts of applications of 
SAS graphics as well as links to his blog, e-books and other graphics information. All of his code can be 
freely downloaded and adapted. 

His code: http://robslink.com/SAS/democd25/timezone.sas is accompanied by the following results 
(http://robslink.com/SAS/democd25/timezone.htm): 

 

 

Figure 2. Time Zone Map from Robslink.com.  

This served as a good starting point and does clearly identify the United States time zones. While 
retaining the general structure, I made a few modifications for usability within the team.  

First, because most of Arizona does not observe daylight savings time, accurately depicting on a map is 
difficult. Two maps could be created – one for when most of the country is observing daylight time and 
one for standard time. The map above highlights the area in Arizona where daylight savings time is 
observed but makes it difficult to know the current time. Since no team members are in Arizona, and since 
Arizona is technically in the Mountain time zone regardless of observance of daylight savings time, I 
included all of Arizona in the Mountain zone. 

Other modifications included: 

 Removing Alaska and Hawaii and their time zones. Since there are no team members located in 
either state, a continental United States map is all that is needed.  

 Adding state labels (postal abbreviations) so states can be more easily identified. 

 Fixing a South Dakota county. Bob's code predates a change in county FIPS code for Oglala 
County, South Dakota and using code as is results in a blank space in South Dakota. 

 Changing the title and title font. 

The figure below shows the time zone map with the above modifications: 

https://communities.sas.com/
http://robslink.com/SAS/Home.htm
http://robslink.com/SAS/democd25/timezone.sas
http://robslink.com/SAS/democd25/timezone.htm

Figure 3. Time Zone Map from Robslink.com with modifications.  

Here is partial code for these modifications (also exclude Alaska, Hawaii, Virgin Islands and Puerto Rico 
throughout): 

/*Fix South Dakota - change county number to match map data*/ 

data mydata; set mydata; if county=102 then county=113; run; 

/*Add state labels*/ 

data maplabel; 

length function $ 8; 

  retain flag 0 xsys ysys '2' hsys '3' when 'a' style "'Albany AMT'"; 

  set maps.uscenter(drop=long lat); 

  where fipstate(state) ^in ('AK','HI','PR','VI','DC');  

  function='label'; text=fipstate(state); size=2.5; position='5'; 

   color='black'; 

  if ocean='Y' then do;    

     position='6'; output;   

     function='move'; flag=1; 

   end;      

  else if flag=1 then do;                  

     function='draw'; size=.5; flag=0; 

   end; 

  output; 

run; 

While the map above provided state information, it still did not provide information about the location of 
team members. Additional modifications included: 

 Removing county boundaries. 

 Selecting lighter colors to create more contrast for state abbreviations and member locations. 

 Marking team member locations with stars and tool-tips for mouse-overs.  

 Since both 'Time Zone' and 'Timezone' are correct according to the dictionary, I used the 
alternate format here. 

Figure 4. Time Zone Map with team member information.  

Here is partial code for the new modifications: 

/*Combine team member locations with cities in mapsgfk.uscity dataset 

for location coordinates*/  

Proc SQL; 

create table cities as select *  

  from team(keep=city state poptext) t, mapsgfk.uscity u  

where t.state=u.state and t.city=u.city; 

quit; 

/*Create annotate data set for star marker and mouse-over text*/ 

data citystar; 

  length function style color $ 8 position $ 1 

    text $ 20 html $ 100; 

  retain xsys ysys "2" hsys "3" when "a"; 

set cities(keep=x y poptext); 

  html= 'alt=' ||trim(left(poptext))||' '; 

  function="symbol"; style="marker"; text="V"; color="red";  

    size=4; position=5; 

output;  

run; 

/*Combine the citystar annotate data set with maplabel annotate set 

(state abbreviations) to use together on GMAP CHORO statement*/ 

data label; length html $ 100; 

set maplabel citystar; 

run; 

/*Map the data*/ 

Proc GMAP data=mydata map=mymap anno=outline all;  

  id state county;  

  choro timezone / legend=legend1 annotate=label coutline=same; 

run; quit; 

SOLUTION TWO: SHAPEFILE TIME ZONES 

Another approach to time zones is importing and using a time zone boundary shapefile. Shapefiles allow 
a boundary to be drawn around the time zones. World time zone boundaries are built using 
OpenStreetMap data as part of a github time zone boundary project. The shapefiles are available for 
download at; https://github.com/evansiroky/timezone-boundary-builder. Currently shapefiles are available 
with or without oceans. Each shape or object has a single attribute in the shapefile called tzid. The tzid 
corresponds to the time zone name as defined in the created time zone database. Here is the shapefile 
for the continental United States mapped with SAS without modifications.  

 

Figure 5. Time Zone Map with imported shapefile.  

In the map above, note the large number of time zone names. Many of these were created when areas 
within the larger time zone did not observe daylight savings time. Although, in most cases, observance is 
universal within the time zone (see information on Arizona above), these delineations still exist. 

Here is the SAS code used to create the time zone map (where combined-shapefile.shp is the 
downloaded shape file. 

/*Import the shapefile*/ 

Proc MAPIMPORT datafile="/mydir/SESUG 2018/Timezone/dist/combined-

shapefile.shp" out=work.tz_shp;  

run;  

 

/*Select continental US*/ 

data tz_us; 

set tz_shp(where=(tzid in ('America/Boise' 'America/Chicago' 

  'America/Denver' 'America/Detroit' 'America/Indiana/Indianapolis' 

  'America/Indiana/Knox' 'America/Indiana/Marengo' 

  'America/Indiana/Petersburg' 'America/Indiana/Tell_City' 

  'America/Indiana/Vevay' 'America/Indiana/Vincennes' 

  'America/Indiana/Winamac' 'America/Kentucky/Louisville' 

  'America/Kentucky/Monticello' 'America/Los_Angeles'  

  'America/Menominee' 'America/New_York' 'America/North_Dakota/Beulah' 

  'America/North_Dakota/Center' 'America/North_Dakota/New_Salem' 

  'America/Phoenix')));  

run; 

 

/*Prepare for projection*/ 

data tzus_shp; 

set tz_us; 

https://github.com/evansiroky/timezone-boundary-builder

    Y= atan(1)/45*Y; 

  X = -(atan(1)/45*X); 

run; 

 

/*Project the data*/ 

Proc GPROJECT data=tzus_shp out=tzus_map; 

  id tzid; 

run; 

 

/*Map the data*/ 

Proc GMAP data=tzus_map map=tzus_map; 

id tzid;  

choro tzid; 

run; quit; 

This map provided the boundary information needed to draw lines around the time zones. To make it 
useful for identifying team member locations, I made the following modifications: 

 Removed non-relevant time zones. SAS Proc REMOVE was used to roll these smaller areas into 
the four larger time zones. 

 Added state boundaries and state abbreviations. The addition of state boundaries and their labels 
(postal abbreviations) helps orient the user. 

 Changed colors. Because I was using several layers of data, I used transparent colors. 

 Added team member locations (with mouse-over) as above. 

 Removed the legend. This is not needed with the clock images. 

 Added clocks. Clocks show the time change by time zone and label the time zones. 

Figure 6 shows the new, customized time zone map created from the shape file and illustrates the 
mouse-over ability. 

 

Figure 6. Customized Time Zone Map from Shape File.  

To create this map, the SAS code uses the annotate data sets illustrated above for state labels, state 
outlines and team member locations with mouse over. Additional annotate data sets are created for the 
time zone boundaries and colors. See code below: 

/*Time Zone Boundary Outline*/ 
data outlinetz; set tzus_shp; 

  by tz SEGMENT notsorted; 

length function color $9; 

  xsys='2'; ysys='2'; when='b'; line=1; size=2; 

  color='A0000FF33'; style='mempty';  

if first.segment then function='poly'; 

  else function='polycont'; 

run; 

 

/*Time Zone Colors*/ 
data outlinetzf; set tzus_shp; 

  by tz SEGMENT notsorted; 

length function color $9; 

  xsys='2'; ysys='2'; when='b'; line=1; size=3; 

  color='AFF000033'; style='S';  

if tz=3 then color='A98FB9833'; 

  else if tz=2 then color='AFFFF0033'; 

  else if tz=4 then color='AADDAE633'; 

  else if tz=1 then color='AF4A46033'; 

if first.segment then function='poly'; 

  else function='polycont';run; 

Clocks are also added with annotate. Placement of the clocks uses percentage of the output area. Clock 
image inclusion code is provided below: 

data zone; set zone_anno; 

length function $8 imgpath $200 ; 

  xsys='3'; ysys='3'; hsys='3'; when='a'; 

if tz=3 then do; 

  Position='B'; 

  function="move"; x=75; y=82; output; 

  function="image"; style="fit";  

  imgpath="/mydir/TZ_E_.jpg"; 

  x=x+7; y=y+9; 

output; end; 

if tz=1 then do; 

Position='B'; 

  function="move"; x=50.5; y=82; output; 

  function="image"; style="fit";  

  imgpath="/mydir/TZ_C_.jpg"; 

  x=x+7; y=y+9; 

output; end; 

if tz=2 then do; 

  Position='B'; 

  function="move"; x=30; y=84; output; 

  function="image"; style="fit";  

  imgpath="/mydir/TZ_M_.jpg"; 

  x=x+7; y=y+9; 

output; end; 

if tz=4 then do; 

  Position='B'; 

  function="move"; x=12.5; y=88; output; 

  function="image"; style="fit"; 

  imgpath="/mydir/TZ_P_.jpg"; 

  x=x+7; y=y+9; 

output; end;  

run; 

The Annotate data sets were combined for mapping since GMAP only allows two annotates – one on the 
GMAP statement and one on the CHORO statement. 

A NOTE CONCERNING OPENSTREETMAP AND TIME ZONE BOUNDARIES: 

OpenStreetMap (OSM) is a free, editable map of the world that is being built and continuously updated by 
volunteers and released with an open-content license. The OpenStreetMap License allows access to 
most map images and all of the underlying map data. At this time, OpenStreetMap has made the decision 
to not create stand-alone boundaries for time zones as part of OSM. Time zone information is available 
as tags for administrative boundaries such as countries, states, territories, counties and cities. These tags 
are what have been used to create the shapefile used in the example above. 

CONCLUSION 

In today’s world of distributed employees and virtual workplaces, it is not enough to know where your 
fellow employees are, it is also important to know their time zone so that meetings and other corporate 
events can be held at times that do not inconvenience a segment of the workforce. As illustrated above, 
SAS is a great solution for visualizing this information. Like most other applications of SAS software, for 
time zone mapping I did not need to 'reinvent the wheel.' Someone had already created a starting point 
for me to create these maps. 

Appendix 1: Time Zones. 

State United States Time Zones 

AL Central 

AR Central 

AS Alaska; Aleutian Islands: Hawaii-Aleutian. 

AZ Mountain; The Navajo Nation uses Daylight Saving Time, the rest of the state does not. 

CA Pacific 

CO Mountain 

CT Eastern 

DC Eastern 

DE Eastern 

FL Most of the state: Eastern; West of the Apalachicola River: Central. 

GA Eastern 

HI Hawaii-Aleutian; State does not use Daylight Savings Time. 

IA Central 

ID Most of the state: Mountain; North of the Salmon River: Pacific. 

IL Central 

IN Most of the state: Eastern; Northwest and southwest corners: Central. 

KS Most of the state: Central; Greeley, Hamilton, Sherman and Wallace counties: Mountain. 

KY Western half of the state: Central; Eastern half of the state: Eastern. 

LA Central 

MA Eastern 

MD Eastern 

ME Eastern 

MI Most of the state: Eastern; Counties that share a border with Wisconsin: Central. 

MN Central 

MO Central 

MS Central 

MT Mountain 

NC Eastern 

ND Most of the state: Central; Southwestern part of the state: Mountain. 

NE Most of the state: Central; Western part of the state: Mountain. 

NH Eastern 

NJ Eastern 

NM Mountain 

NV Most of the state: Pacific; Towns of Jackpot and West Wendover: Mountain. 

NY Eastern 

OH Eastern 

OK Central 

OR Most of the state: Pacific; Part of Malheur County: Mountain. 

PA Eastern 

RI Eastern 

SC Eastern 

SD Eastern half of the state: Central; Western half of the state: Mountain. 

TN 
West Tennessee and Middle Tennessee plus Marion County: Central; East Tennessee except 
Marion County: Eastern. 

TX 
Most of the state: Central; El Paso and Hudspeth counties and part of Culberson County: 
Mountain. 

UT Mountain 

VA Eastern 

VT Eastern 

WA Pacific 

WI Central 

WV Eastern 

WY Mountain 

Table 1. Time Zones of the United States. 

 


